# Maintainer: Lukas Fleischer <lfleischer@archlinux.org>
# Contributor: Daniel Wallace <danielwallace at gtmanfred dot com>
# Contributor: Army

pkgname=vim-latexsuite
_pkgname=vim-latex
pkgver=1.10.0
pkgrel=4
epoch=1
pkgdesc="Tools to view, edit, and compile LaTeX documents within Vim."
arch=('any')
url="http://vim-latex.sourceforge.net"
license=('custom:vim')
depends=('perl' 'vim-plugin-runtime')
optdepends=('python' 'texlive-bin')
groups=('vim-plugins')
source=("https://sourceforge.net/projects/${_pkgname}/files/releases/${_pkgname}-${pkgver}.tar.gz"{,.asc})
sha256sums=('f1a9d2d0446c99d9fa0fa83386609ee5201a0de3f1e7ff4aae8d40763ed84f6d'
            'SKIP')
validpgpkeys=('18A0E3D6A36194E0A6F2C5F06A3A10B31C109517') # Till Maas
install=$pkgname.install

package() {
	cd "${srcdir}/${_pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" PREFIX='/usr' VIMDIR='/usr/share/vim/vimfiles' install
}
